clear all
set more off
set type double
set seed 20260227

* ==================== Path Settings ====================
* IMPORTANT: Ensure there are no Chinese characters in your file path
[bookmark: _GoBack]global project "C:/Users/Lenovo/Desktop/Aging_Employment/MyResearch"
cd "$project"

cap mkdir "$project/data/cleaned"
cap mkdir "$project/temp"
cap mkdir "$project/output"

* Install required packages
local req_cmds "reghdfe estout ftools center winsor2 ivreghdfe xtoverid"
foreach cmd of local req_cmds {
    cap which `cmd'
    if _rc ssc install `cmd', replace
}

* ==================== 1. Data Import ====================
import excel "WDI_panel_data.xlsx", sheet("Sheet1") firstrow clear

* Standardize country and year variable names
cap rename CountryName country
cap rename countryname country
cap rename Country country
cap rename COUNTRY country
cap rename Year year
cap rename YEAR year

* ==================== 2. Variable Renaming (English) =====================
* Dependent Variables: Youth Unemployment Rates
rename SLUEM1524ZS    YUR_total
rename SLUEM1524MAZS  YUR_male
rename SLUEM1524FEZS  YUR_female

* Robustness Dependent Variables: Youth NEET Rates
rename SLUEMNEETMEZS    NEET_total
rename SLUEMNEETMAMEZS  NEET_male
rename SLUEMNEETFEMEZS  NEET_female

* Core Independent Variables
rename SPPOPDPNDOL    Aging          /* Old-age dependency ratio */
rename SPPOP65UPTOZS  Aging_65up     /* Population ages 65 and above (% of total) */

* Moderating Variable
rename NECONTOTLZS    Consume        /* Final consumption expenditure (% of GDP) */

* Control Variables
rename NYGDPMKTPKDZG  GDP_growth
rename FPCPITOTLZG    Inflation      /* CPI */
rename SPPOPGROW      Pop_growth
rename SPURBTOTLINZS  Urbanization
rename NETRDGNFSZS    Trade_openness /* FDI / Trade openness */

* ==================== 3. String to Numeric Conversion ====================
local str_vars YUR_total YUR_male YUR_female ///
               NEET_total NEET_male NEET_female ///
               Aging Aging_65up ///
               Consume ///
               GDP_growth Inflation Pop_growth Urbanization Trade_openness

foreach v of local str_vars {
    cap confirm string variable `v'
    if !_rc {
        replace `v' = "" if `v' == ".."
        destring `v', replace force
    }
}

* ==================== 4. Country Encoding ====================
gen country_clean = trim(lower(country))
encode country_clean, gen(country_code)
drop country_clean
xtset country_code year, yearly

* ==================== 5. Process Main Sample ====================
keep if year >= 2005 & year <= 2025

* ==================== 6. Exclude Countries/Regions with Pop < 1 Million ====================
gen exclude_flag = 0

* [Exclusion list for small economies]
replace exclude_flag = 1 if regexm(lower(country), "american samoa")
replace exclude_flag = 1 if regexm(lower(country), "andorra")
replace exclude_flag = 1 if regexm(lower(country), "antigua")
replace exclude_flag = 1 if regexm(lower(country), "aruba")
replace exclude_flag = 1 if regexm(lower(country), "bahamas")
replace exclude_flag = 1 if regexm(lower(country), "barbados")
replace exclude_flag = 1 if regexm(lower(country), "belize")
replace exclude_flag = 1 if regexm(lower(country), "bermuda")
replace exclude_flag = 1 if regexm(lower(country), "bhutan")
replace exclude_flag = 1 if regexm(lower(country), "brunei")
replace exclude_flag = 1 if regexm(lower(country), "cabo verde") | regexm(lower(country), "cape verde")
replace exclude_flag = 1 if regexm(lower(country), "cayman")
replace exclude_flag = 1 if regexm(lower(country), "channel islands")
replace exclude_flag = 1 if regexm(lower(country), "macao") | regexm(lower(country), "macau")
replace exclude_flag = 1 if regexm(lower(country), "comoros")
replace exclude_flag = 1 if regexm(lower(country), "curacao") | regexm(lower(country), "curaçao")
replace exclude_flag = 1 if regexm(lower(country), "dominica") & !regexm(lower(country), "dominican")
replace exclude_flag = 1 if regexm(lower(country), "faroe")
replace exclude_flag = 1 if regexm(lower(country), "fiji")
replace exclude_flag = 1 if regexm(lower(country), "french polynesia")
replace exclude_flag = 1 if regexm(lower(country), "greenland")
replace exclude_flag = 1 if regexm(lower(country), "grenada")
replace exclude_flag = 1 if regexm(lower(country), "guam")
replace exclude_flag = 1 if regexm(lower(country), "guyana")
replace exclude_flag = 1 if regexm(lower(country), "iceland")
replace exclude_flag = 1 if regexm(lower(country), "isle of man")
replace exclude_flag = 1 if regexm(lower(country), "kiribati")
replace exclude_flag = 1 if regexm(lower(country), "liechtenstein")
replace exclude_flag = 1 if regexm(lower(country), "luxembourg")
replace exclude_flag = 1 if regexm(lower(country), "maldives")
replace exclude_flag = 1 if regexm(lower(country), "malta")
replace exclude_flag = 1 if regexm(lower(country), "marshall islands")
replace exclude_flag = 1 if regexm(lower(country), "micronesia")
replace exclude_flag = 1 if regexm(lower(country), "monaco")
replace exclude_flag = 1 if regexm(lower(country), "montenegro")
replace exclude_flag = 1 if regexm(lower(country), "new caledonia")
replace exclude_flag = 1 if regexm(lower(country), "northern mariana")
replace exclude_flag = 1 if regexm(lower(country), "palau")
replace exclude_flag = 1 if regexm(lower(country), "samoa") & !regexm(lower(country), "american")
replace exclude_flag = 1 if regexm(lower(country), "san marino")
replace exclude_flag = 1 if regexm(lower(country), "sao tome") | regexm(lower(country), "são tomé")
replace exclude_flag = 1 if regexm(lower(country), "seychelles")
replace exclude_flag = 1 if regexm(lower(country), "sint maarten")
replace exclude_flag = 1 if regexm(lower(country), "solomon islands")
replace exclude_flag = 1 if regexm(lower(country), "st. kitts") | regexm(lower(country), "saint kitts")
replace exclude_flag = 1 if regexm(lower(country), "st. martin") | regexm(lower(country), "saint martin")
replace exclude_flag = 1 if regexm(lower(country), "st. vincent") | regexm(lower(country), "saint vincent")
replace exclude_flag = 1 if regexm(lower(country), "suriname")
replace exclude_flag = 1 if regexm(lower(country), "tonga")
replace exclude_flag = 1 if regexm(lower(country), "turks and caicos")
replace exclude_flag = 1 if regexm(lower(country), "tuvalu")
replace exclude_flag = 1 if regexm(lower(country), "vanuatu")
replace exclude_flag = 1 if regexm(lower(country), "virgin islands") & regexm(lower(country), "u.s.")

di "========== Excluded Countries (Population < 1 million) =========="
tab country if exclude_flag == 1

* Exclude these countries
drop if exclude_flag == 1
drop exclude_flag

di "========== Sample Size After Exclusion =========="
count

* ==================== 7. Winsorization (1% and 99%) ====================
local winsor_vars YUR_total YUR_male YUR_female ///
                  NEET_total NEET_male NEET_female ///
                  Aging Aging_65up ///
                  Consume ///
                  GDP_growth Inflation Pop_growth Urbanization Trade_openness

foreach v of local winsor_vars {
    winsor2 `v', cuts(1 99) replace
}

* ==================== 8. Construct Base Sample ====================
global controls "GDP_growth Inflation Pop_growth Urbanization Trade_openness"

* Generate sample marker based on non-missing core variables
egen missing_count = rowmiss(YUR_total YUR_male YUR_female ///
                             Aging Aging_65up Consume ///
                             $controls)

gen base_sample = (missing_count == 0)
tab base_sample, m

* Keep only base sample
keep if base_sample == 1
drop missing_count base_sample

* Drop countries with less than 5 years of observations (ensure panel balance)
bysort country_code: gen N_country = _N
drop if N_country < 5
tab N_country        
count                

di "========== Sample Size After Removing Short Panels =========="
count

* ==================== 9. Mean Centering (For Interaction Terms) =====================
center Aging Consume, prefix(c_)

* Generate interaction term
gen c_Aging_Consume = c_Aging * c_Consume
label var c_Aging_Consume "Aging * Consumption (Centered)"

* ==================== 10. Save Cleaned Sample ====================
log using "$project/output/sample_selection.log", replace text
di "Original Observations: 4,522"
di "Excluded pop < 1M: 1,087 obs (54 countries)"
di "Missing core variables: 767 obs"
di "Short panels removed: 7 obs"
di "Final Sample Size: 2,661 obs"
log close

save "$project/data/cleaned/full_sample.dta", replace
di "========== Base sample saved. Total obs = " _N "=========="


* ==================== Hausman Test ====================
di "========== Hausman Test: FE vs RE =========="

use "$project/data/cleaned/full_sample.dta", clear
global controls "GDP_growth Inflation Pop_growth Urbanization Trade_openness"

* Step 1: Fixed Effects Model (FE)
xtreg YUR_total Aging $controls, fe
estimates store FE

* Step 2: Random Effects Model (RE)
xtreg YUR_total Aging $controls, re
estimates store RE

* Step 3: Traditional Hausman Test
log using "$project/output/hausman_test.log", replace text
di "========== Hausman Test Results =========="
hausman FE RE, sigmamore
log close

* Step 4: Robust Hausman Test (xtoverid)
qui xtreg YUR_total Aging $controls, re
xtoverid

* ==================== Table 1: Descriptive Statistics ====================
di "========== Table 1 Descriptive Statistics =========="

* Get regression sample
reghdfe YUR_total Aging $controls, absorb(country_code year) vce(cluster country_code)
gen reg_sample = e(sample)
keep if reg_sample == 1

local desc_vars "YUR_total YUR_male YUR_female Aging Aging_65up Consume GDP_growth Inflation Pop_growth Urbanization Trade_openness"

estpost sum `desc_vars', detail
esttab using "$project/output/Table1_Descriptive_Statistics.rtf", ///
    replace cells("count mean sd min max p50") ///
    title("Table 1: Descriptive Statistics of Core Variables") nonumber nomtitle

* ==================== Table 2: Correlation Matrix ====================
di "========== Table 2 Correlation Matrix =========="
local core_vars "YUR_total Aging Consume GDP_growth Inflation Pop_growth Urbanization Trade_openness"
estpost correlate `core_vars', matrix listwise
esttab using "$project/output/Table2_Correlation_Matrix.rtf", ///
    replace unstack not noobs compress ///
    title("Table 2: Correlation Matrix")

* ==================== Table 3: VIF Test ====================
di "========== Table 3 VIF Test =========="
reg YUR_total Aging Consume $controls
estat vif
matrix vif_matrix = r(vif)
esttab matrix(vif_matrix) using "$project/output/Table3_VIF_Test.rtf", ///
    replace title("Table 3: Variance Inflation Factor (VIF) Test") ///
    addnotes("Note: Mean VIF < 10 indicates no severe multicollinearity.")

* ============================================================================
* Table 4: Baseline Regression
* ============================================================================

* Step 1: Controls only
reghdfe YUR_total $controls Consume, absorb(country_code year) vce(cluster country_code)
est store step1_total

reghdfe YUR_male $controls Consume, absorb(country_code year) vce(cluster country_code)
est store step1_male

reghdfe YUR_female $controls Consume, absorb(country_code year) vce(cluster country_code)
est store step1_female

* Step 2: Independent variable only
reghdfe YUR_total Aging, absorb(country_code year) vce(cluster country_code)
est store step2_total

reghdfe YUR_male Aging, absorb(country_code year) vce(cluster country_code)
est store step2_male

reghdfe YUR_female Aging, absorb(country_code year) vce(cluster country_code)
est store step2_female

* Step 3: Full model
reghdfe YUR_total Aging $controls Consume, absorb(country_code year) vce(cluster country_code)
est store step3_total

reghdfe YUR_male Aging $controls Consume, absorb(country_code year) vce(cluster country_code)
est store step3_male

reghdfe YUR_female Aging $controls Consume, absorb(country_code year) vce(cluster country_code)
est store step3_female

* Export Table 4
esttab step1_total step1_male step1_female ///
       step2_total step2_male step2_female ///
       step3_total step3_male step3_female ///
    using "$project/output/Table4_Baseline_Regression.rtf", ///
    replace ///
    b(3) se(3) star(* 0.1 ** 0.05 *** 0.01) ///
    title("Table 4: Baseline Regression - Impact of Aging on Youth Unemployment") ///
    mgroups("Controls Only" "IV Only" "Full Model", ///
            pattern(1 1 1 1 1 1 1 1 1) ///
            prefix(\multicolumn{@span}{c}{) suffix(}) span) ///
    mtitles("(1)" "(2)" "(3)" "(4)" "(5)" "(6)" "(7)" "(8)" "(9)") ///
    keep(Aging Consume GDP_growth Inflation Pop_growth Urbanization Trade_openness _cons) ///
    order(Aging Consume) ///
    varlabels( ///
        Aging "Aging" ///
        Consume "Consumption" ///
        GDP_growth "GDP Growth" ///
        Inflation "Inflation" ///
        Pop_growth "Pop Growth" ///
        Urbanization "Urbanization" ///
        Trade_openness "Trade Openness" ///
        _cons "Constant") ///
    stats(N r2 N_clust, ///
          fmt(0 3 0) ///
          labels("Observations" "R-squared" "Number of Countries")) ///
    addnotes("Note: Dependent variable is Youth Unemployment Rate. Robust standard errors clustered at the country level are in parentheses. *** p<0.01, ** p<0.05, * p<0.1.") ///
    compress noabbrev

* ==================== Table 5: Moderating Effect (Consumption) ====================
di "========== Table 5 Moderating Effect =========="

reghdfe YUR_total c_Aging c_Consume c_Aging_Consume $controls, ///
    absorb(country_code year) vce(cluster country_code)
est store mod_total

reghdfe YUR_male c_Aging c_Consume c_Aging_Consume $controls, ///
    absorb(country_code year) vce(cluster country_code)
est store mod_male

reghdfe YUR_female c_Aging c_Consume c_Aging_Consume $controls, ///
    absorb(country_code year) vce(cluster country_code)
est store mod_female

esttab mod_total mod_male mod_female ///
    using "$project/output/Table5_Moderating_Effect.rtf", ///
    replace b(3) se(3) star(* 0.1 ** 0.05 *** 0.01) ///
    title("Table 5: Moderating Effect of Consumption on Aging and Youth Unemployment") ///
    mtitles("Total" "Male" "Female") ///
    scalars("r2 R-squared" "N Observations" "N_clust Number of Countries") ///
    addnotes("Note: Interaction variables are mean-centered. Country and year fixed effects are included. Robust standard errors clustered at the country level are in parentheses. *** p<0.01, ** p<0.05, * p<0.1.") ///
    keep(c_Aging c_Consume c_Aging_Consume GDP_growth Inflation Pop_growth Urbanization Trade_openness) ///
    varlabels( ///
        c_Aging "Aging" ///
        c_Consume "Consumption" ///
        c_Aging_Consume "Aging x Consumption" ///
        GDP_growth "GDP Growth" ///
        Inflation "Inflation" ///
        Pop_growth "Pop Growth" ///
        Urbanization "Urbanization" ///
        Trade_openness "Trade Openness" ///
    ) width(\hsize) nonumber

* ==================== Heterogeneity: Income Group Classification ====================
import excel "$project/income_classification.xlsx", sheet("List of economies") clear

rename A country_name
rename B iso3
rename C region
rename D income_level
rename E income_group
rename F lending_category

drop if country_name == "country"
destring income_group, replace
drop if missing(country_name)

gen str50 country_match = trim(country_name)
duplicates drop country_match, force
isid country_match
save "$project/temp/income_classification.dta", replace

use "$project/data/cleaned/full_sample.dta", clear
gen str50 country_match = trim(country)
replace country_match = "Côte d'Ivoire" if country == "Cote d'Ivoire"
replace country_match = "Vietnam" if country == "Viet Nam"

merge m:1 country_match using "$project/temp/income_classification.dta"
drop if _merge == 2

* Manual matching for missing countries
replace income_group = 2 if country == "Cote d'Ivoire" & _merge==1
replace income_level = "Lower middle income" if country == "Cote d'Ivoire" & _merge==1
replace income_group = 1 if country == "Ethiopia" & _merge==1
replace income_level = "Low income" if country == "Ethiopia" & _merge==1
replace income_group = 3 if country == "Turkiye" & _merge==1
replace income_level = "Upper middle income" if country == "Turkiye" & _merge==1
replace income_group = 3 if country == "Venezuela, RB" & _merge==1
replace income_level = "Upper middle income" if country == "Venezuela, RB" & _merge==1
drop _merge

* Generate developed/developing dummies
gen developed = (income_group == 4)
gen developing = (income_group >= 1 & income_group <= 3)

label variable developed "Developed Countries"
label variable developing "Developing Countries"
save "$project/data/cleaned/full_sample_with_income.dta", replace

* ==================== Table 6: Heterogeneity Analysis (Developed vs Developing) ====================
di "========== Table 6 Heterogeneity: Developed vs Developing =========="

preserve
keep if developed == 1
reghdfe YUR_total c_Aging $controls, absorb(country_code year) vce(cluster country_code)
est store dev_base
reghdfe YUR_total c_Aging c_Consume c_Aging_Consume $controls, absorb(country_code year) vce(cluster country_code)
est store dev_cons
restore

preserve
keep if developing == 1
reghdfe YUR_total c_Aging $controls, absorb(country_code year) vce(cluster country_code)
est store undev_base
reghdfe YUR_total c_Aging c_Consume c_Aging_Consume $controls, absorb(country_code year) vce(cluster country_code)
est store undev_cons
restore

esttab dev_base undev_base dev_cons undev_cons ///
    using "$project/output/Table6_Heterogeneity_Income.rtf", ///
    replace b(3) se(3) star(* 0.1 ** 0.05 *** 0.01) ///
    title("Table 6: Heterogeneity Analysis - Developed vs Developing Countries") ///
    mtitles("Developed" "Developing" "Developed" "Developing") ///
    mgroups("Baseline" "Moderation", pattern(1 0 1 0)) ///
    order(c_Aging c_Aging_Consume c_Consume) ///
    varlabels( ///
        c_Aging "Aging" ///
        c_Aging_Consume "Aging x Consumption" ///
        c_Consume "Consumption" ///
        GDP_growth "GDP Growth" ///
        Inflation "Inflation" ///
        Pop_growth "Pop Growth" ///
        Urbanization "Urbanization" ///
        Trade_openness "Trade Openness") ///
    drop(_cons) ///
    scalars("r2 R-squared" "N Observations" "N_clust Number of Countries") ///
    addnotes("Note: Dependent variable is Youth Unemployment Rate. Country and year fixed effects are included. Robust standard errors clustered at the country level are in parentheses. *** p<0.01, ** p<0.05, * p<0.1") ///
    width(\hsize) nonumber

* ==================== Heterogeneity: Service Sector Employment ====================
reghdfe YUR_total Aging $controls, absorb(country_code year) vce(cluster country_code)
gen base_sample = e(sample)

cap confirm variable SLSRVEMPLZS
if _rc {
    di as error "Variable SLSRVEMPLZS (Service employment %) not found."
    exit 111
}

sum SLSRVEMPLZS if base_sample==1, detail
gen high_serv_emp = (SLSRVEMPLZS > r(p50)) if base_sample==1 & !missing(SLSRVEMPLZS)

reghdfe YUR_total c_Aging $controls if high_serv_emp==0 & base_sample==1, absorb(country_code year) vce(cluster country_code)
est store low_serv_base
reghdfe YUR_total c_Aging c_Consume c_Aging_Consume $controls if high_serv_emp==0 & base_sample==1, absorb(country_code year) vce(cluster country_code)
est store low_serv_mod

reghdfe YUR_total c_Aging $controls if high_serv_emp==1 & base_sample==1, absorb(country_code year) vce(cluster country_code)
est store high_serv_base
reghdfe YUR_total c_Aging c_Consume c_Aging_Consume $controls if high_serv_emp==1 & base_sample==1, absorb(country_code year) vce(cluster country_code)
est store high_serv_mod

esttab low_serv_base high_serv_base low_serv_mod high_serv_mod ///
    using "$project/output/Table7_Heterogeneity_Service.rtf", ///
    replace b(3) se(3) star(* 0.1 ** 0.05 *** 0.01) ///
    title("Table 7: Heterogeneity Analysis by Service Sector Employment Share") ///
    mtitles("Low Service" "High Service" "Low Service" "High Service") ///
    order(c_Aging c_Aging_Consume c_Consume) ///
    varlabels( ///
        c_Aging "Aging" ///
        c_Aging_Consume "Aging x Consumption" ///
        c_Consume "Consumption" ///
        GDP_growth "GDP Growth" ///
        Inflation "Inflation" ///
        Pop_growth "Pop Growth" ///
        Urbanization "Urbanization" ///
        Trade_openness "Trade Openness") ///
    drop(_cons) ///
    stats(N r2, fmt(0 3) labels("Observations" "R-squared")) ///
    addnotes("Note: Grouped by median service employment share. Country and year fixed effects are included. *** p<0.01, ** p<0.05, * p<0.1.") ///
    width(\hsize) nonumber

* ============================================================================
* Robustness Checks: Substituting Variables & Trim
* ============================================================================
use "$project/data/cleaned/full_sample.dta", clear
global controls "GDP_growth Inflation Pop_growth Urbanization Trade_openness"

reghdfe YUR_total Aging $controls, absorb(country_code year) vce(cluster country_code)
gen base_sample = e(sample)

* Part 1: Baseline
reghdfe YUR_total Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store base_total
reghdfe YUR_male Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store base_male
reghdfe YUR_female Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store base_female

* Part 2: Replace Independent Variable (Aging 65+)
reghdfe YUR_total Aging_65up $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store replaceX_total
reghdfe YUR_male Aging_65up $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store replaceX_male
reghdfe YUR_female Aging_65up $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store replaceX_female

* Part 3: Replace Dependent Variable (NEET Rate)
reghdfe NEET_total Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store replaceY_total
reghdfe NEET_male Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store replaceY_male
reghdfe NEET_female Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store replaceY_female

* Part 4: 5% Winsorization
foreach var of varlist YUR_total YUR_male YUR_female NEET_total NEET_male NEET_female Aging Aging_65up GDP_growth Inflation Pop_growth Urbanization Trade_openness {
    gen `var'_orig = `var'
}
winsor2 YUR_total YUR_male YUR_female NEET_total NEET_male NEET_female Aging Aging_65up GDP_growth Inflation Pop_growth Urbanization Trade_openness, cuts(2.5 97.5) replace

reghdfe YUR_total Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store winsor_total
reghdfe YUR_male Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store winsor_male
reghdfe YUR_female Aging $controls if base_sample==1, absorb(country_code year) vce(cluster country_code)
est store winsor_female

* Export Robustness Table
esttab base_total base_male base_female ///
       replaceX_total replaceX_male replaceX_female ///
       replaceY_total replaceY_male replaceY_female ///
       winsor_total winsor_male winsor_female ///
    using "$project/output/Table8_Robustness_Checks.rtf", ///
    replace b(3) se(3) star(* 0.1 ** 0.05 *** 0.01) ///
    title("Table 8: Robustness Checks") ///
    mtitles("Total" "Male" "Female" "Total" "Male" "Female" "Total" "Male" "Female" "Total" "Male" "Female") ///
    mgroups("Baseline" "Replace IV" "Replace DV" "5% Trim", pattern(1 0 0 1 0 0 1 0 0 1 0 0)) ///
    keep(Aging Aging_65up) ///
    varlabels(Aging "Aging" Aging_65up "Aging 65+") ///
    stats(N r2 N_clust, fmt(0 3 0) labels("Observations" "R-squared" "Number of Countries")) ///
    addnotes("Note: Dependent variable for 'Replace DV' columns is NEET Rate. Country and year fixed effects are included. *** p<0.01, ** p<0.05, * p<0.1.") ///
    width(\hsize) nonumber

* ============================================================================
* Instrumental Variables (IV - 2SLS)
* ============================================================================
use "$project/data/cleaned/full_sample.dta", clear
global controls "GDP_growth Inflation Pop_growth Urbanization Trade_openness"

sort country_code year
by country_code: gen Aging_lag1 = Aging[_n-1] if year == year[_n-1] + 1
by country_code: gen Aging_lag2 = Aging[_n-2] if year == year[_n-2] + 2

reghdfe YUR_total Aging $controls, absorb(country_code year) vce(cluster country_code)
gen base_sample = e(sample)

* Total Sample - Lag 1 & 2
ivreghdfe YUR_total (Aging = Aging_lag1) $controls if base_sample==1, absorb(country_code year) cluster(country_code) first
est store iv_total_lag1
ivreghdfe YUR_total (Aging = Aging_lag2) $controls if base_sample==1, absorb(country_code year) cluster(country_code) first
est store iv_total_lag2

* Male Sample - Lag 1 & 2
ivreghdfe YUR_male (Aging = Aging_lag1) $controls if base_sample==1, absorb(country_code year) cluster(country_code) first
est store iv_male_lag1
ivreghdfe YUR_male (Aging = Aging_lag2) $controls if base_sample==1, absorb(country_code year) cluster(country_code) first
est store iv_male_lag2

* Female Sample - Lag 1 & 2
ivreghdfe YUR_female (Aging = Aging_lag1) $controls if base_sample==1, absorb(country_code year) cluster(country_code) first
est store iv_female_lag1
ivreghdfe YUR_female (Aging = Aging_lag2) $controls if base_sample==1, absorb(country_code year) cluster(country_code) first
est store iv_female_lag2

* Export IV Table
esttab iv_total_lag1 iv_total_lag2 iv_male_lag1 iv_male_lag2 iv_female_lag1 iv_female_lag2 ///
    using "$project/output/Table9_IV_Estimation.rtf", ///
    replace b(3) se(3) star(* 0.1 ** 0.05 *** 0.01) ///
    title("Table 9: Instrumental Variable (IV) Estimation using Lagged Aging") ///
    mtitles("Total L1" "Total L2" "Male L1" "Male L2" "Female L1" "Female L2") ///
    mgroups("Total" "Male" "Female", pattern(1 0 1 0 1 0)) ///
    keep(Aging $controls) ///
    varlabels( ///
        Aging "Aging" ///
        GDP_growth "GDP Growth" ///
        Inflation "Inflation" ///
        Pop_growth "Pop Growth" ///
        Urbanization "Urbanization" ///
        Trade_openness "Trade Openness") ///
    stats(N N_clust r2 rkf, fmt(0 0 3 2) labels("Observations" "Number of Countries" "R-squared" "KP F-Statistic")) ///
    addnotes("Note: Dependent variable is Youth Unemployment Rate. Instrumental variables are the first lag (L1) and second lag (L2) of Aging. KP F-statistic denotes the Kleibergen-Paap rk Wald F-statistic for weak instrument test. *** p<0.01, ** p<0.05, * p<0.1.") ///
    width(\hsize) nonumber
